Maple Workbook of Calculus Problems
- Preliminary Version -

Bruce Char
Bev Marshman
Ian McGee
Peter Ponzo

Research Report CS-87-27
April 1987

INTRODUCTION

Since the Spring of 1985 some of us at the University of Waterloo have been considering the problems involved in integrating symbolic computation (i.e. Maple) into the undergraduate mathematics curriculum. Although there are numerous obstacles to their effective use (not the least of which is convincing faculty that it's a good idea!) a major obstacle is the lack of "suitable" problems which teach mathematics while having the student use a symbolic computation system.

Symbolic computation gives us the opportunity to assign problems which (normally) involve long/tedious calculations. The student is asked to describe a procedure for its solution, using Maple to perform the computations, thereby testing the validity of the procedure. The student can experiment, make conjectures based upon a number of special cases, discover some interesting property ... without being concerned with (or inhibited by) the details of the computation.

Another way to look at the use of symbolic computation is that the student is free to sharpen his/her high-level problem-solving skills by relying on Maple to handle the mundane calculation details. Students discover that the crucial part of certain problems is deciding upon the mathematical statements (i.e. formulas, constraints) that accurately model the problem situation, and the sequence of formal operations (differentiation, indefinite integration, solution of equations) that will derive the formulas, pictures, or numbers that provide the desired information.

A recent grant from the Alfred P. Sloan Foundation has, as one of its objectives, the compilation of a WORKBOOK of "suitable" problems. Although our Sloan project runs until August, 1988, we felt it may be useful to others (and might provide us with much-needed feedback) if we mailed our problems as they are generated ... hence this (initial) collection.

It is expected that the final WORKBOOK will include a collection of "problem-sets". Each problem-set would have a "sample", completely worked out (with Maple commands and response), to illustrate the nature of the set and make transparent the necessary Maple syntax (it it our intention to obviate, as much as possible, the need to learn a computer language!). The problem-set would then continue with related problems for the students (having the problem statement only). The student would be expected to describe (as Maple #comments) each step in his/her solution. The problems in this collection have the same flavour, with 'problem.0' intended as a sample and the remaining problems in a set intended as assignment problems.

Insofar as is possible, the problems (in the final WORKBOOK) will be multi-step, multi-concept problems organized according to the mathematical concepts involved.

We intend, also, to produce an accompanying instructor's manual for the WORKBOOK, which would contain full MAPLE solutions to all the problems, and possibly some indication of the pedagogical goals of each set. (This collection has full Maple solutions).

Note: A second project (under the Sloan grant) is to write software which simplifies the creation of "tutorials" in mathematics. The student may (at a terminal) type 'display taylor-series' and a program called 'display' will read a text file written by a faculty member (in this case 'taylor-series') which provides a lesson on some topic in mathematics (or other subject). The text file includes directives for the 'display' program such as "Plot a Graph" or "Ask a Question" or "Ask Maple for the Derivative of some Function", etc. (The 'Plots' in these "problem-sets" were produced by 'displaying' such a text file).

All problems (as they are created), as well as the 'display' program (which, in its current state, runs on a VAX and requires a vt240 terminal), are in a "sloan guest_account" on a VAX at Waterloo (available to all interested users). If you wish to login, to extract problems, please write to any of the below (and ask for the userid/password) or mail to 'watsloan@watmum.bitnet'. We would also appreciate any comments/criticisms/ etc., hearing of your own experiences, and problem suggestions.

Electronic mail to sloan-forum%watnot@waterloo.csnet will be automatically sent to some of the other institutions participating in the Sloan program (and may be used as a vehicle for discussion).

Much of the computer equipment used for this project was made available via the "Waterloo-Digital Equipment Joint Research Agreement".

The Maple software and/or manual may be obtained by writing to :

WATCOM, 415 Phillip St., Waterloo, Ontario, Canada

Bruce Char, Department of computer Science (bwchar@watmum)
Bev Marshman, Department of Applied Mathematics (bjmarshman@watbun)
Ian McGee , Department of Applied Mathematics (ijmcgee@watmath)
Peter Ponzo, Department of Applied Mathematics (pjponzo@watmath)
University of Waterloo,
Waterloo, Ontario,
N2L 3G1

GEOMETRY OF CUBIC CURVE PROPERTY



GEOMETRY OF CUBIC CURVE PROPERTY



```
#          PROBLEMS on a CUBIC CURVE PROPERTY
#          ==============================
# PROBLEM 1.0:
# ===========
# Here, a problem in introductory calculus which involves the calculation
# of tangent lines, solving non-linear equatuions, and computing areas.
#  CUBIC CURVE PROPERTY:
#        Suppose P is any point on the graph of y=x^3. The tangent at P
# crosses the curve at Q, and A is the area between the curve and
# the line segment PQ. Similarly, the tangent at Q meets the curve again
# at R, and B is the area between the curve and QR.
#        Show that B is always 16 times as great as A for every choice
# of the point P.
# (adapted from problem 380, Crux Mathematicorum, 1979, page 171).
# (Try:  "display 16X.0.Plot"  for graphical display).
# ========================================================================
# Define the curve:
> y:=x^3;
                                         3
                                   y := x

-----------------------------------------------------------------------------
# Assume the co-ordinates of the point P are (k,k^3),say. Then the tangent
# line PQ has equation:
> PQ:=k^3+3*k^2*(x-k);
                                3      2
                         PQ := k  + 3 k  (x - k)

-----------------------------------------------------------------------------
# Determine the x-coordinate of intersection of PQ and the curve:
> solve(PQ=y,x);
                               - 2 k, k, k

-----------------------------------------------------------------------------
# The double root 'k' is the point of tangency. The other is the x-coordinate
# of the point Q.
# Calculate the area A:
> areaA:=int(PQ-y,x=k..-2*k);
                                             4
                              areaA := 27/4 k

-----------------------------------------------------------------------------
# Write the equation of the tangent line QR:
> QR:=-8*k^3+12*k^2*(x+2*k);
                                 3        2
                      QR := - 8 k  + 12 k  (x + 2 k)

-----------------------------------------------------------------------------
# Find the x-coordinate of R:
> solve(y=QR,x);
                            4 k, - 2 k, - 2 k

-----------------------------------------------------------------------------
# Now calculate the area B:
> areaB:=int(y-QR,x=4*k..-2*k);
                                            4
                              areaB := 108 k

-----------------------------------------------------------------------------
# Compare areas;
> evalf(areaB/areaA);
                                    16.

-----------------------------------------------------------------------------
```

```
# PROBLEM 1.1:
# ===========
# In problem 1.0, it is shown that the property B = 16A is true for
# any point P on the curve y = x^3. Show that this same property is true
# for the curve y = x^(1/3). This function is the 'inverse' function
# of y = x^3, so that it should be clear that the property holds.
# ======================================================================
# Define the curve:
> y:=x^(1/3);
```

$$y := x^{1/3}$$

```
# Assume the co-ordinates of the point P are [k,k^(1/3)],say. Then the
# tangent line PQ has equation:
> PQ:=k^(1/3)+(1/3)*k^(-2/3)*(x-k);
```

$$PQ := k^{1/3} + 1/3\ \frac{x - k}{k^{2/3}}$$

```
# Determine the x-coordinate of intersection of PQ and the curve:
> solve(PQ=y,x);
```

$$- 8\ k,\ k,\ k$$

```
# The double root 'k' is the point of tangency. The other is the x-coordinate
# of the point Q.
# Calculate the area A:
> areaA:=int(PQ-y,x=k..-8*k);
```

$$areaA := 21/4\ k^{4/3} - 3/4\ (-8)^{4/3}\ k^{4/3}$$

```
# Write the equation of the tangent line QR:
> QR:=-2*k^(1/3)+(1/12)*k^(-2/3)*(x+8*k);
```

$$QR := - 2\ k^{1/3} + 1/12\ \frac{x + 8\ k}{k^{2/3}}$$

```
# Find the x-coordinate of R:
> solve(y=QR,x);
```

$$64\ k,\ - 8\ k,\ - 8\ k$$

```
# Now calculate the area B:
> areaB:=int(y-QR,x=-8*k..64*k);
```

$$areaB := 3/4\ 64^{4/3}\ k^{4/3} - 72\ k^{4/3} - 3/4\ (-8)^{4/3}\ k^{4/3}$$

```
# Compare areas;
> evalf(areaB/areaA);
```

$$7.652173913$$

```
# PROBLEM 1.2:
# ==========
# In problem 1.0, it is shown that the property B = 16A is true for
# any point P on the curve y = x^3. Show that this same property is true
# for the every curve of the third degree. Use y = ax^3 + bx^2 + cx + d,
# and the coordinates of the point P [x1,y1].
# (Try:  "display 16X.2.Plot"  for graphical display).
# ======================================================================
# Define the curve:
> y:=a*x^3+b*x^2+c*x+d;
```

$$y := a\,x^3 + b\,x^2 + c\,x + d$$

```
# The slope of the line PQ is to be the same as that of the curve at x1:
> slope:=subs(x=x1,diff(",x));
```

$$slope := 3\,a\,x1^2 + 2\,b\,x1 + c$$

```
# and y1 is given in terms of x1 by:
> y1:=subs(x=x1,y);
```

$$y1 := a\,x1^3 + b\,x1^2 + c\,x1 + d$$

```
# Then the tangent line PQ has equation:
> PQ:=y1+slope*(x-x1);
```

$$PQ := a\,x1^3 + b\,x1^2 + c\,x1 + d + (3\,a\,x1^2 + 2\,b\,x1 + c)\,(x - x1)$$

```
# Determine the x-coordinate of intersection of PQ and the curve:
> solns:=solve(PQ=y,x);
```

$$solns := x1,\ x1,\ -\frac{2\,a\,x1 + b}{a}$$

```
# The double root 'x1' is the point of tangency. The other is the x-coordinate
# of the point Q.
> x2:=solns[3];
```

$$x2 := -\frac{2\,a\,x1 + b}{a}$$

```
# Calculate the area A:
> areaA:=int(PQ-y,x=x1..x2);
```

$$areaA := -x1^3\,(2\,a\,x1 + b) - \frac{b\,x1^2\,(2\,a\,x1 + b)}{a} - \frac{c\,x1\,(2\,a\,x1 + b)}{a}$$

$$+ (3\,a\,x1^2 + 2\,b\,x1 + c)\left(1/2\,\frac{(2\,a\,x1 + b)^2}{a^2} + \frac{x1\,(2\,a\,x1 + b)}{a}\right)$$

$$- 1/4\,\frac{(2\,a\,x1 + b)^4}{a^3} + 1/3\,\frac{b\,(2\,a\,x1 + b)^3}{a^3} - 1/2\,\frac{c\,(2\,a\,x1 + b)^2}{a^2}$$

$$- 3/4\,a\,x1^4 - 2/3\,b\,x1^3 - 1/2\,c\,x1^2 + 1/2\,(3\,a\,x1^2 + 2\,b\,x1 + c)\,x1^2$$

```
# Now determine the equation of the tangent line QR:
> slope:=subs(x=x2,diff(y,x));
> y2:=subs(x=x2,y);
> QR:=y2+slope*(x-x2);
```

$$slope := 3\,\frac{(2\,a\,x1 + b)^2}{a} - 2\,\frac{b\,(2\,a\,x1 + b)}{a} + c$$

$$y2 := -\frac{(2\,a\,x1 + b)^3}{a^2} + \frac{b\,(2\,a\,x1 + b)^2}{a^2} - \frac{c\,(2\,a\,x1 + b)}{a} + d$$

$$QR := -\frac{(2\,a\,x1 + b)^3}{a^2} + \frac{b\,(2\,a\,x1 + b)^2}{a^2} - \frac{c\,(2\,a\,x1 + b)}{a} + d + \left(3\,\frac{(2\,a\,x1 + b)^2}{a} - 2\,\frac{b\,(2\,a\,x1 + b)}{a} + c\right)\left(x + \frac{2\,a\,x1 + b}{a}\right)$$

```
# Find the x-coordinate of R:
> solns:=solve(y=QR,x);
```

$$solns := \frac{b + 4\,a\,x1}{a},\ -\frac{2\,a\,x1 + b}{a},\ -\frac{2\,a\,x1 + b}{a}$$

```
# Now calculate the area B:
> areaB:=int(y-QR,x=solns[1]..x2);
```

$$areaB := -3/4\,\frac{(2\,a\,x1 + b)^4}{a^3} + 2/3\,\frac{b\,(2\,a\,x1 + b)^3}{a^3} - 1/2\,\frac{c\,(2\,a\,x1 + b)^2}{a^2} + 1/2\,\frac{\left(3\,\frac{(2\,a\,x1 + b)^2}{a} - 2\,\frac{b\,(2\,a\,x1 + b)}{a} + c\right)(2\,a\,x1 + b)^2}{a^2} - 1/4\,\frac{(b + 4\,a\,x1)^4}{a^3} - 1/3\,\frac{b\,(b + 4\,a\,x1)^3}{a^3} - 1/2\,\frac{c\,(b + 4\,a\,x1)^2}{a^2} - \frac{(2\,a\,x1 + b)^3\,(b + 4\,a\,x1)}{a^3} + \frac{b\,(2\,a\,x1 + b)^2\,(b + 4\,a\,x1)}{a^3}$$

```
     c  (2  a  x1  +  b)   (b  +  4  a  x1)
   - ------------------------------
                           2
                          a

                              2
      (2  a  x1  +  b)              b  (2  a  x1  +  b)
   +  (3  ---------------  -  2  ----------------  +  c)
              a                          a

                                   2
        (b  +  4  a  x1)              (2  a  x1  +  b)   (b  +  4  a  x1)
    (1/2  ---------------  +  ---------------------------)
                    2                                 2
                   a                                 a

---------------------------------------------------------------------------
#===============================================================
# Now compare areas;
> A:=expand(areaA);
> B:=expand(areaB);
                                       2    2                     3          4
                            4         b   x1              3      b   x1     b
       A  :=  27/4  a  x1   +  9/2  --------  +  9  b  x1   +  -------  +  1/12  ----
                                       a                          2         3
                                                                 a         a

                                    2    2                          3         4
                          4        b   x1                 3        b   x1     b
      B  :=  108  a  x1   +  72  --------  +  144  b  x1   +  16  -------  +  4/3  ----
                                    a                                 2           3
                                                                     a           a

---------------------------------------------------------------------------
# PROBLEM 1.3:
# ==========
# Perform the same sequence of operations as in Problem 1.0, but on
# the function y = ax^(1/3) + bx. Compute the ratio of area A and area B.
# =================================================================
# Define the curve:
> y:=a*x^(1/3)+b*x;
                                       1/3
                               y  :=  a  x     +  b  x
---------------------------------------------------------------------------
# Assume the co-ordinates of the point P are [x1,y1]. Then the
# slope of the line PQ is to be the same as that of the curve at x1:
> slope:=subs(x=x1,diff(",x));
                                          a
                            slope  :=  1/3  -------  +  b
                                           2/3
                                         x1
---------------------------------------------------------------------------
# and y1 is given in terms of x1 by:
> y1:=subs(x=x1,y);
                                         1/3
                              y1  :=  a  x1     +  b  x1

---------------------------------------------------------------------------
# Then the tangent line PQ has equation:
> PQ:=y1+slope*(x-x1);
                         1/3                             a
          PQ  :=  a  x1     +  b  x1  +  (1/3  -------  +  b)   (x  -  x1)
                                                    2/3
                                                  x1
```

```
--------------------------------------------------------------------------------
# Determine the x-coordinate of intersection of PQ and the curve:
> solns:=solve(PQ=y,x);
                          solns := - 8 x1, x1, x1

--------------------------------------------------------------------------------
# The double root 'x1' is the point of tangency. The other is the x-coordinate
# of the point Q.
> x2:=solns[1];
                                x2 := - 8 x1

--------------------------------------------------------------------------------
# Calculate the area A:
> areaA:=int(PQ-y,x=x1..x2);
                      4/3                   4/3                4/3
     areaA := - 1/12 a x1    (- 64 + 9 (-8)   ) - 1/12 a x1

--------------------------------------------------------------------------------
# Now determine the equation of the tangent line QR:
> slope:=subs(x=x2,diff(y,x));
> y2:=subs(x=x2,y);
> QR:=y2+slope*(x-x2);
                                      a
                     slope := 1/3 -------------- + b
                                      2/3   2/3
                                  (-8)    x1

                                       1/3   1/3
                      y2 := a (-8)    x1     - 8 b x1

               1/3   1/3                          a
   QR := a (-8)    x1    - 8 b x1 + (1/3 -------------- + b) (x + 8 x1)
                                             2/3   2/3
                                         (-8)    x1


--------------------------------------------------------------------------------
# Find the x-coordinate of R:
> soln:=solve(y=QR,x);
                      solns := 64 x1, - 8 x1, - 8 x1

--------------------------------------------------------------------------------
# Now calculate the area B:
> areaB:=int(y-QR,x=solns[1]..x2);
                           4/3               4/3      4/3     2/3
                       a x1               a x1    (9 64    (-8)    - 4096)
     areaB := - 16/3 --------- - 1/12 -----------------------------------
                           2/3                             2/3
                       (-8)                            (-8)

--------------------------------------------------------------------------------
#===============================================================
# Now compare areas;
> A:=expand(areaA);
> B:=expand(areaB);
                             4/3             4/3   4/3
                A := 21/4 a x1    - 3/4 (-8)    x1    a

                              4/3
                          a x1               4/3   4/3
                B := 336 --------- - 3/4 64    x1    a
                              2/3
                          (-8)
--------------------------------------------------------------------------------
```

GAUSS QUADRATURE PROBLEMS

```
#
# PROBLEM 1.0:
# ===========
# Assume y(x) is a cubic. Choose constants A, B and points u, v on (-1,1)
# so that the integral of y(x), from x=-1 to x=1 is obtained EXACTLY from
# the sum: A y(u) + B y(v).
# Use your result on:
```

$$y = p x^3 + q x^2 + r x + s, \quad y = x^4, \quad y = \sin(x), \quad y = \cos(x), \quad y = \exp(-x)$$

```
# ======================================================================
# Define a general cubic
> y:=a*x^3+b*x^2+c*x+d;
```

$$y := a x^3 + b x^2 + c x + d$$

```
# Compute integral from x=-1 to x=1
> INTEGRAL:=int(y,x=-1..1);
```

$$INTEGRAL := 2/3\ b + 2\ d$$

```
# Define points y1=y(u), y2=y(v)
> y1:=subs(x=u,y);
> y2:=subs(x=v,y);
```

$$y1 := a u^3 + b u^2 + c u + d$$

$$y2 := a v^3 + b v^2 + c v + d$$

```
# Compute sum Ay1+By2
> SUM:=A*y1+B*y2;
```

$$SUM := A\ (a u^3 + b u^2 + c u + d) + B\ (a v^3 + b v^2 + c v + d)$$

```
# Equate coefficients of "a", "b", "c", "d", in SUM and INTEGRAL
> EQN1:=diff(SUM,a)=diff(INTEGRAL,a);
> EQN2:=diff(SUM,b)=diff(INTEGRAL,b);
> EQN3:=diff(SUM,c)=diff(INTEGRAL,c);
> EQN4:=diff(SUM,d)=diff(INTEGRAL,d);
```

$$EQN1 := A u^3 + B v^3 = 0$$

$$EQN2 := A u^2 + B v^2 = 2/3$$

$$EQN3 := A u + B v = 0$$

$$EQN4 := A + B = 2$$

```
# We solve these 4 equations for A, B, u and v :
# First solve EQN4 for A
> A:=solve(EQN4,A);
```

$$A := - B + 2$$

```
# Next solve EQN3 for B
> B:=solve(EQN3,B);
```

$$B := - 2 \frac{u}{- u + v}$$

```
------------------------------------------------------------------------------
# Simplify the expressions
> A:=factor(A);
> B:=factor(B);
                                            v
                              A := - 2 -------
                                          u - v

                                          u
                               B := 2 -------
                                        u - v

------------------------------------------------------------------------------
# The remaining two equations are
> EQN1:=factor(EQN1);
> EQN2:=factor(EQN2);
                        EQN1 := - 2 u v (u + v) = 0

                           EQN2 := - 2 u v = 2/3

------------------------------------------------------------------------------
# Since neither u nor v can be zero (else EQN2 is not satisfied)
# then u=-v, from EQN1
> u:=-v;
                                  u := - v

------------------------------------------------------------------------------
# Finally, solve EQN2 for v
> v:=solve(EQN2,v);
                                  1         1
                         v := ------, - ------
                                1/2       1/2
                               3         3

------------------------------------------------------------------------------
# Choose the positive root for v
> v:=1/sqrt(3);
                                       1
                                 v := ------
                                        1/2
                                       3

------------------------------------------------------------------------------
# The values of A, B, u and v are:
> A:=A;
> B:=B;
> u:=u;
> v:=v;
                                  A := 1

                                  B := 1

                                          1
                               u := - ------
                                          1/2
                                         3

                                        1
                                 v := ------
                                         1/2
                                        3

------------------------------------------------------------------------------
```

```
# The final result is then:
#
#      x=1                       1               1
#  INTEGRAL f(x)dx = f( --- ) + f( - --- ), exact for ALL cubic f(x).
#     x=-1                       /3              /3
#
# Try it on a cubic
> y:=p*x^3+q*x^2+r*x+s;
> INTEGRAL:=int(y,x=-1..1);
> SUM:=subs(x=-1/sqrt(3),y)+subs(x=1/sqrt(3),y);
                                        3       2
                           y := p x  + q x  + r x + s

                             INTEGRAL := 2/3 q + 2 s

                               SUM := 2/3 q + 2 s

------------------------------------------------------------------------------
# Try y = x^4
> y:=x^4;
> INTEGRAL:=int(y,x=-1..1);
> SUM:=subs(x=-1/sqrt(3),y)+subs(x=1/sqrt(3),y);
                                          4
                                    y := x

                                INTEGRAL := 2/5

                                   SUM := 2/9

------------------------------------------------------------------------------
# Try y = sin(x)
> y:=sin(x);
> INTEGRAL:=evalf(int(y,x=-1..1));
> SUM:=evalf( subs(x=-1/sqrt(3),y)+subs(x=1/sqrt(3),y) );
                                  y := sin(x)

                                 INTEGRAL := 0

                                    SUM := 0

------------------------------------------------------------------------------
# Try y = cos(x)
> y:=cos(x);
> INTEGRAL:=evalf(int(y,x=-1..1));
> SUM:=evalf( subs(x=-1/sqrt(3),y)+subs(x=1/sqrt(3),y) );
                                  y := cos(x)

                            INTEGRAL := 1.682941970

                               SUM := 1.675823656

------------------------------------------------------------------------------
# Try y = exp(-x)
> y:=exp(-x);
> INTEGRAL:=evalf(int(y,x=-1..1));
> SUM:=evalf( subs(x=-1/sqrt(3),y)+subs(x=1/sqrt(3),y) );
                                 y := exp(- x)

                            INTEGRAL := 2.350402387

                               SUM := 2.342696088

------------------------------------------------------------------------------
```

```
# PROBLEM 1.1:
# ===========
# Assume y(x) is a quintic. Choose constants A, B, C and points u, v, w
# on (-1,1) so that the integral of y(x), from x=-1 to x=1 is obtained exactly
# from the sum: A y(u) + B y(v) + C y(w). (Hint: try for a solution with
# the points u, v, w symmetrically located about x=0).
# Use your formula for:
```

$$y=Px^5 + Q x^4 + R x^3 + S x^2 + T x + U,\ y=x^6,\ y=\sin(x),\ y=\cos(x),\ y=\exp(-x)$$

```
# ====================================================================
# Define a general quintic
> y:=a*x^5+b*x^4+c*x^3+d*x^2+e*x+f;
```

$$y := a\,x^5 + b\,x^4 + c\,x^3 + d\,x^2 + e\,x + f$$

```
# Compute integral from x=-1 to x=1
> INTEGRAL:=int(y,x=-1..1);
```

$$INTEGRAL := 2/5\ b + 2/3\ d + 2\ f$$

```
# Define points y1=y(u), y2=y(v), y3=y(w)
> y1:=subs(x=u,y);
> y2:=subs(x=v,y);
> y3:=subs(x=w,y);
```

$$y1 := a\,u^5 + b\,u^4 + c\,u^3 + d\,u^2 + e\,u + f$$

$$y2 := a\,v^5 + b\,v^4 + c\,v^3 + d\,v^2 + e\,v + f$$

$$y3 := a\,w^5 + b\,w^4 + c\,w^3 + d\,w^2 + e\,w + f$$

```
# Compute sum Ay1+By2+Cy3
> SUM:=A*y1+B*y2+C*y3;
```

$$SUM := A\,(a\,u^5 + b\,u^4 + c\,u^3 + d\,u^2 + e\,u + f)$$

$$+ B\,(a\,v^5 + b\,v^4 + c\,v^3 + d\,v^2 + e\,v + f)$$

$$+ C\,(a\,w^5 + b\,w^4 + c\,w^3 + d\,w^2 + e\,w + f)$$

```
# Equate coefficients of "a", "b", "c" .... "f", in SUM and INTEGRAL
> EQN1:=diff(SUM,a)=diff(INTEGRAL,a);
> EQN2:=diff(SUM,b)=diff(INTEGRAL,b);
> EQN3:=diff(SUM,c)=diff(INTEGRAL,c);
> EQN4:=diff(SUM,d)=diff(INTEGRAL,d);
> EQN5:=diff(SUM,e)=diff(INTEGRAL,e);
> EQN6:=diff(SUM,f)=diff(INTEGRAL,f);
```

$$EQN1 := A\,u^5 + B\,v^5 + C\,w^5 = 0$$

$$EQN2 := A\,u^4 + B\,v^4 + C\,w^4 = 2/5$$

$$EQN3 := A\,u^3 + B\,v^3 + C\,w^3 = 0$$

$$EQN4 := A\,u^2 + B\,v^2 + C\,w^2 = 2/3$$

$$EQN5 := A\,u + B\,v + C\,w = 0$$

$$EQN6 := A + B + C = 2$$

```
# We solve these equations for A, B, C, u, v and w :
# First solve EQN6 for A
> A:=solve(EQN6,A);
```

$$A := -B - C + 2$$

```
# Next solve EQN5 for B
> B:=solve(EQN5,B);
```

$$B := -\frac{-C u + 2 u + C w}{-u + v}$$

```
# Next solve EQN4 for C
> C:=solve(EQN4,C);
```

$$C := -\frac{2 \frac{u^3}{-u + v} + 2 u^2 - 2 \frac{u v^2}{-u + v} - 2/3}{-\frac{u^3}{-u + v} + \frac{w u^2}{-u + v} - u^2 + \frac{u v^2}{-u + v} - \frac{w v^2}{-u + v} + w^2}$$

```
# Simplify the expressions (retaining some symmetry)
> A:=factor(A);
> B:=factor(B);
> C:=factor(C);
```

$$A := 2/3 \frac{3 v w + 1}{(u - w)\ (u - v)}$$

$$B := -2/3 \frac{3 u w + 1}{(-w + v)\ (u - v)}$$

$$C := 2/3 \frac{3 u v + 1}{(-w + v)\ (u - w)}$$

```
# Try for a solution with v=0 (seeking points u, v, w symmetrical about x=0)
> v:=0;
```

$$v := 0$$

```
# The remaining equations become
> EQN1:=factor(EQN1);
> EQN2:=factor(EQN2);
> EQN3:=factor(EQN3);
```

$$EQN1 := 2/3\ (u + w)\ (u^2 + w^2) = 0$$

$$EQN2 := 2/3\ u^2 + 2/3\ u w + 2/3\ w^2 = 2/5$$

$$EQN3 := 2/3\ u + 2/3\ w = 0$$

```
# Choose u=-w (satisfying EQN1 and EQN3); points -w, 0, w ARE symmetrical.
> u:=-w;
```

$$u := -w$$

```
---------------------------------------------------------------------------
# Solve EQN2 for w
> w:=solve(EQN2,w);
                                  1            1
                        w := 3 -------, - 3 -------
                                 1/2          1/2
                               15           15

---------------------------------------------------------------------------
# Choose positive root
> w:=3/sqrt(15);
                                         1
                              w := 3 -------
                                       1/2
                                     15

---------------------------------------------------------------------------
# The values of A, B, C, u, v and w are:
> A:=A; B:=B; C:=C; u:=u; v:=v; w:=w;
                                A := 5/9

                                B := 8/9

                                C := 5/9

                                          1
                              u := - 3 -------
                                         1/2
                                       15

                                 v := 0

                                         1
                              w := 3 -------
                                       1/2
                                     15

---------------------------------------------------------------------------
#
# The final result (valid for ALL polynomials of degree < 6) is:
#
#     x=1                  5       3       8          5      3
#   INTEGRAL  f(x)dx =  --- f(- ---)  +  --- f(0)  +  --- f(---)
#     x=-1                 9     /15       9          9    /15
#
# Try a quintic
> y:=P*x^5+Q*x^4+R*x^3+S*x^2+T*x+U;
> INTEGRAL:=int(y,x=-1..1);
> SUM:=A*subs(x=u,y)+B*subs(x=v,y)+C*subs(x=w,y);
                        5       4       3       2
             y := P x  + Q x  + R x  + S x  + T x + U

                INTEGRAL := 2/5 Q + 2/3 S + 2 U

                   SUM := 2/5 Q + 2/3 S + 2 U

---------------------------------------------------------------------------
```

```
# Try y=x^6
> y:=x^6;
> INTEGRAL:=evalf( int(y,x=-1..1) );
> SUM:=evalf( A*subs(x=u,y)+B*subs(x=v,y)+C*subs(x=w,y) );
                                      6
                                y := x

                     INTEGRAL := .2857142857

                         SUM := .2400000000

--------------------------------------------------------------------------
# Try y=sin(x)
> y:=sin(x);
> INTEGRAL:=evalf( int(y,x=-1..1) );
> SUM:=evalf( A*subs(x=u,y)+B*subs(x=v,y)+C*subs(x=w,y) );
                              y := sin(x)

                           INTEGRAL := 0

                               SUM := 0

--------------------------------------------------------------------------
# Try y=cos(x)
> y:=cos(x);
> INTEGRAL:=evalf( int(y,x=-1..1) );
> SUM:=evalf( A*subs(x=u,y)+B*subs(x=v,y)+C*subs(x=w,y) );
                              y := cos(x)

                     INTEGRAL := 1.682941970

                         SUM := 1.683003547

--------------------------------------------------------------------------
# Try y=exp(-x)
> y:=exp(-x);
> INTEGRAL:=evalf( int(y,x=-1..1) );
> SUM:=evalf( A*subs(x=u,y)+B*subs(x=v,y)+C*subs(x=w,y) );
                             y := exp(- x)

                     INTEGRAL := 2.350402387

                         SUM := 2.350336929

--------------------------------------------------------------------------
```

PROBLEM 1.2:

```
# ==========
# Assume y(x) is a 7th degree polynomial. Choose constants A, B and points u, v
# on (0,1) so that the integral of y(x), from x=-1 to x=1 is obtained exactly
# from the sum: A y(-u) + B y(-v) + B y(v) + A y(u).
# Use your formula for:
#     7     6     5     4     3     2                 8
# y=ax + b x + c x + d x + e x + f x + g x + h,  y = x ,
# y=sin(x), y=cos(x), y=exp(-x)
# ==============================================================================
# Define a 7th degree polynomial
> y:=a*x^7+b*x^6+c*x^5+d*x^4+e*x^3+f*x^2+g*x+h;
```

$$y := a\,x^7 + b\,x^6 + c\,x^5 + d\,x^4 + e\,x^3 + f\,x^2 + g\,x + h$$

```
# Compute integral from x=-1 to x=1
> INTEGRAL:=int(y,x=-1..1);
```

$$INTEGRAL := 2/7\ b + 2/5\ d + 2/3\ f + 2\ h$$

```
# Define points y1=y(-u), y2=y(-v), y3=y(v) and y4=y(u)
> y1:=subs(x=-u,y);
> y2:=subs(x=-v,y);
> y3:=subs(x=v,y);
> y4:=subs(x=u,y);
```

$$y1 := -a\,u^7 + b\,u^6 - c\,u^5 + d\,u^4 - e\,u^3 + f\,u^2 - g\,u + h$$

$$y2 := -a\,v^7 + b\,v^6 - c\,v^5 + d\,v^4 - e\,v^3 + f\,v^2 - g\,v + h$$

$$y3 := a\,v^7 + b\,v^6 + c\,v^5 + d\,v^4 + e\,v^3 + f\,v^2 + g\,v + h$$

$$y4 := a\,u^7 + b\,u^6 + c\,u^5 + d\,u^4 + e\,u^3 + f\,u^2 + g\,u + h$$

```
# Compute sum Ay1+By2+By3+Ay4
> SUM:=A*y1+B*y2+B*y3+A*y4;
```

$$SUM := A\,(-a\,u^7 + b\,u^6 - c\,u^5 + d\,u^4 - e\,u^3 + f\,u^2 - g\,u + h)$$

$$+ B\,(-a\,v^7 + b\,v^6 - c\,v^5 + d\,v^4 - e\,v^3 + f\,v^2 - g\,v + h)$$

$$+ B\,(a\,v^7 + b\,v^6 + c\,v^5 + d\,v^4 + e\,v^3 + f\,v^2 + g\,v + h)$$

$$+ A\,(a\,u^7 + b\,u^6 + c\,u^5 + d\,u^4 + e\,u^3 + f\,u^2 + g\,u + h)$$

```
# Equate coefficients of "a", "b", "c" .... "h", in SUM and INTEGRAL
> EQN1:=diff(SUM,a)=diff(INTEGRAL,a);
> EQN2:=diff(SUM,b)=diff(INTEGRAL,b);
> EQN3:=diff(SUM,c)=diff(INTEGRAL,c);
> EQN4:=diff(SUM,d)=diff(INTEGRAL,d);
> EQN5:=diff(SUM,e)=diff(INTEGRAL,e);
> EQN6:=diff(SUM,f)=diff(INTEGRAL,f);
> EQN7:=diff(SUM,g)=diff(INTEGRAL,g);
> EQN8:=diff(SUM,h)=diff(INTEGRAL,h);
```

$$EQN1 := 0 = 0$$

$$EQN2 := 2\,A\,u^6 + 2\,B\,v^6 = 2/7$$

$$EQN3 := 0 = 0$$

$$EQN4 := 2\,A\,u^4 + 2\,B\,v^4 = 2/5$$

$$EQN5 := 0 = 0$$

$$EQN6 := 2\,A\,u^2 + 2\,B\,v^2 = 2/3$$

$$EQN7 := 0 = 0$$

$$EQN8 := 2\,A + 2\,B = 2$$

```
# Half the equations are automatically satisfied (because of the symmetry).
# We solve EQN2, EQN4, EQN6 and EQN8 for A, B, u and v:
# First solve EQN8 for A
> A:=solve(EQN8,A);
```

$$A := -B + 1$$

```
# Next solve EQN6 for B
> B:=solve(EQN6,B);
```

$$B := -\frac{2\,u^2 - 2/3}{-2\,u^2 + 2\,v^2}$$

```
# The remaining equations become
> Eqn1:=factor(EQN2);
> Eqn2:=factor(EQN4);
```

$$Eqn1 := 2/3\,u^4 - 2\,v^2\,u^4 + 2/3\,v^2\,u^2 - 2\,v^4\,u^2 + 2/3\,v^4 = 2/7$$

$$Eqn2 := 2/3\,u^2 - 2\,v^2\,u^2 + 2/3\,v^2 = 2/5$$

```
# Since only squares of u and v occur, substitute u=sqrt(U) and v=sqrt(V)
> u:=sqrt(U); v:=sqrt(V);
```

$$u := U^{1/2}$$

$$v := V^{1/2}$$

```
# Equations become
> Eqn1:=factor(Eqn1); Eqn2:=factor(Eqn2);
```

$$Eqn1 := 2/3\,U^2 - 2\,V\,U^2 + 2/3\,V\,U - 2\,V^2\,U + 2/3\,V^2 = 2/7$$

$$Eqn2 := 2/3\,U - 2\,V\,U + 2/3\,V = 2/5$$

```
# Solve Eqn2 for V
> V:=solve(Eqn2,V);
```

$$V := -\frac{2/3\,U - 2/5}{-2\,U + 2/3}$$

```
# Solve Eqn1 for U
> Eqn1:=factor(Eqn1);
> solve(Eqn1,U);
```

$$Eqn1 := 2/75\,\frac{20\,U^2 + 15\,U - 9}{3\,U - 1} = 2/7$$

$$3/7 + 6/7\,\frac{2^{1/2}}{15^{1/2}},\; 3/7 - 6/7\,\frac{2^{1/2}}{15^{1/2}}$$

```
# Choose the larger root (the smaller root is V !!)
> U:=3/7+6/7*sqrt(2/15);
```

$$U := 3/7 + 6/7\,(2/15)^{1/2}$$

```
# Then A, B and u, v are:
> A:=simplify(expand(A));
> B:=simplify(expand(B)): B:=simplify(expand(B));
> u:=simplify(expand(u));
> v:=simplify(expand(v));
```

$$A := 49\,\frac{1}{108 + 6\,30^{1/2}}$$

$$B := \frac{59 + 6\,30^{1/2}}{108 + 6\,30^{1/2}}$$

$$u := \frac{3^{1/2}\left(1 + \frac{2^{3/2}}{15^{1/2}}\right)^{1/2}}{7^{1/2}}$$

$$v := \frac{(-\,15^{1/2} + 5\,2^{1/2})^{1/2}\,3^{1/2}}{5^{1/2}\,(15^{1/2} + 9\,2^{1/2})^{1/2}}$$

```
# Evaluating as a decimal (introducing some round-off error but simplifying !)
> A:=evalf(A); B:=evalf(B); u:=evalf(u); v:=evalf(v);
```

$$A := .3478548450$$

```
                    B := .6521451546

                    u := .8611363118

                    v := .3399810434
--------------------------------------------------------------------------------
#
# We now have, for ALL polynomials of degree < 8,
#
#   x=1
# INTEGRAL f(x)dx = A [f(-u) + f(u)] + B [f(-v) + f(v)]
#  x=-1
#
# Try general 7th degree polynomial
> y:=a*x^7+b*x^6+c*x^5+d*x^4+e*x^3+f*x^2+g*x+h;
> INTEGRAL:=evalf( int(y,x=-1..1) );
> SUM:=evalf( A*(subs(x=-u,y)+subs(x=u,y)) + B*(subs(x=-v,y)+subs(x=v,y)) );
                 7       6       5       4       3       2
        y := a x  + b x  + c x  + d x  + e x  + f x  + g x + h

    INTEGRAL := .2857142857 b + .4000000000 d + .6666666667 f + 2. h

    SUM := .2857142860 b + .4000000000 d + .6666666665 f + 1.999999999 h
--------------------------------------------------------------------------------
# Try
> y:=x^8;
> INTEGRAL:=evalf( int(y,x=-1..1) );
> SUM:=evalf( A*(subs(x=-u,y)+subs(x=u,y)) + B*(subs(x=-v,y)+subs(x=v,y)) );
                                    8
                              y := x

                      INTEGRAL := .2222222222

                         SUM := .2106122452
--------------------------------------------------------------------------------
# Try
> y:=sin(x);
> INTEGRAL:=evalf( int(y,x=-1..1) );
> SUM:=evalf( A*(subs(x=-u,y)+subs(x=u,y)) + B*(subs(x=-v,y)+subs(x=v,y)) );
                            y := sin(x)

                           INTEGRAL := 0

                              SUM := 0
--------------------------------------------------------------------------------
# Try
> y:=cos(x);
> INTEGRAL:=evalf( int(y,x=-1..1) );
> SUM:=evalf( A*(subs(x=-u,y)+subs(x=u,y)) + B*(subs(x=-v,y)+subs(x=v,y)) );
                            y := cos(x)

                      INTEGRAL := 1.682941970

                         SUM := 1.682941688
--------------------------------------------------------------------------------
# Try
> y:=exp(-x);
> INTEGRAL:=evalf( int(y,x=-1..1) );
> SUM:=evalf( A*(subs(x=-u,y)+subs(x=u,y)) + B*(subs(x=-v,y)+subs(x=v,y)) );
                           y := exp(- x)

                      INTEGRAL := 2.350402387

                         SUM := 2.350402092
--------------------------------------------------------------------------------
```

Landing Problems



```
#                           LANDING PROBLEMS
#                           ================
# Problem 1.0
# ===========
# The design of a ground-controlled automatic landing system
# calls for a landing approach to a runway as shown:
#
```


```
#
# The altitude is "h" when descent commences, and a constant
# horizontal airspeed "-U" is maintained.
# In addition the maximum absolute vertical acceleration
# allowed is "g/10".
# Find a cubic polynomial y = f(x) which gives an acceptable
# trajectory.                                          2
# If U = 150m.p.h., h = 1 mile, g = 32 ft/sec , find the
# minimum value of R when descent should commence.
# (Try: "display Landing.Plots"   for  graphical display).
#==============================================================
#-The trajectory is of the form:
> y:=a*x^3+b*x^2+c*x+d;
> yp:=diff(y,x);
```

$$y := a\,x^3 + b\,x^2 + c\,x + d$$

$$yp := 3\,a\,x^2 + 2\,b\,x + c$$

```
#-For smooth landing at the runway we require y(0) = 0, and y'(0) = 0.
# Hence d = 0, and c = 0.
> d:=0:c:=0:
```

```
#-For smooth onset of descent we require y(R) = h, and y'(R) = 0.
#-These conditions can be used to solve for a and b:
> b:=solve(subs(x=R,y)=h,b);
> a:=solve(subs(x=R,yp)=0,a);
```

$$b := -\frac{a\,R^3 - h}{R^2}$$

$$a := -2\,\frac{h}{R^3}$$

```
#-This gives the trajectory in terms of h and R:
> y:=simplify(y);
```

$$y := -2\,\frac{h\,x^3}{R^3} + 3\,\frac{h\,x^2}{R^2}$$

```
# Now calculate the vertical acceleration:
# dy/dt = (dy/dx)(dx/dt) = -U(dy/dx).
# Thus d2y/dt2 = U^2(d2y/dx2).
> Vaccel:=U^2*factor(diff(yp,x));
```

$$Vaccel := -6\,\frac{U^2\,h\,(2\,x - R)}{R^3}$$

```
# The maximum vertical acceleration occurs at the end points:
# x = 0, and x = R, where its magnitude is:
> subs(x=0,");
```

$$6\,\frac{U^2\,h}{R^2}$$

```
# We thus require:          6h(U/R)^2 <= g/10,
# which sets the minimum distance R:  Rmin <= U*sqrt(60h/g).
#                  ==========================
# For U = 150mph, h = 1 mile, g = 32 ft/sec^2, we obtain the distance
# in miles:

> Rmin:=evalf(150*sqrt(60*5280/32)/3600);
```

Rmin := 4.145780988

```
# Problem 1.1
# ==========
# The cubic landing approach of problem.0 has a discontinuity
# in the vertical acceleration at the initial and final "end"
# points. This would cause a "jerk". [Similarly if one brakes
# a car hard until a standstill, there is a jerk as the car
# stops - a discontinuity in the acceleration]. To make the
# acceleration continuous, one needs y''(0) = 0 and y''(R) = 0,
# Redo the landing approach problem with a 5th degree polynomial.
#=================================================================
#-The trajectory is of the form:
> y:=a*x^5+b*x^4+c*x^3+d*x^2+e*x+f;
> yp:=diff(y,x);
```

$$y := a\,x^5 + b\,x^4 + c\,x^3 + d\,x^2 + e\,x + f$$

$$yp := 5\,a\,x^4 + 4\,b\,x^3 + 3\,c\,x^2 + 2\,d\,x + e$$

```
#-For smooth landing at the runway we require y(0) = 0, and y'(0) = 0.
# Hence f = 0, and e = 0.
> f:=0:e:=0:
```

```
#-For smooth onset of descent we require y(R) = h, and y'(R) = 0.
#-These conditions can be used to solve for c and d:
> d:=solve(subs(x=R,y)=h,d);
> c:=solve(subs(x=R,yp)=0,c);
```

$$d := - \frac{a\,R^5 + b\,R^4 + c\,R^3 - h}{R^2}$$

$$c := - \frac{3\,a\,R^4 + 2\,b\,R^3 + 2\,h/R}{R^2}$$

```
#-Finally, to make the acceleration continuous at the end points
# we set y''(0) = 0 and y''(R) = 0 to solve for a and b:
> ypp:=diff(yp,x);
> b:=solve(subs(x=0,ypp)=0,b);
> a:=solve(subs(x=R,ypp)=0,a);
```

$$ypp := 20\,a\,x^3 + 12\,b\,x^2 - 6\,\frac{(3\,a\,R^4 + 2\,b\,R^3 + 2\,h/R)\,x}{R^2}$$

$$- 2\,\frac{a\,R^5 + b\,R^4 - (3\,a\,R^4 + 2\,b\,R^3 + 2\,h/R)\,R - h}{R^2}$$

$$b := \frac{- 2\,a\,R^5 - 3\,h}{R^4}$$

$$a := 6\,\frac{h}{R^5}$$

#-This gives the trajectory in terms of h and R:

```
> y:=simplify(y);
```

$$y := 6 \frac{h x^5}{R^5} - 15 \frac{h x^4}{R^4} + 10 \frac{h x^3}{R^3}$$

========================

#-Now the vertical acceleration is given by:

```
> Vaccel:=U^2*factor(ypp);
```

$$Vaccel := 60 \frac{U^2 x h (2 x - R) (x - R)}{R^5}$$

#-The maximum vertical acceleration occurs at xm say, where

```
> xm:=factor(solve(diff(ypp,x)=0,x));
> Vaccelmax:=evalf(subs(x=xm,Vaccel));
```

$$xm := 1/6 R (3 + 3^{1/2})$$

$$Vaccelmax := - 5.773502685 \frac{U^2 h}{R^2}$$

#-We thus require: Vaccelmax <= g/10,
which sets the minimum distance R: Rmin = U*sqrt(57.7h/g)
#
#-For U = 150mph, h = 1 mile, g = 32 ft/sec^2, this has the value
(in miles):

```
> Rmin:=evalf(150*sqrt(57.7*5280/32)/3600);
```

$$Rmin := 4.065543732$$

TILE.PROBLEM CURVES

```
Tile.0 { y1=-.5773502690*x*(-.535898384*x*x + 1.803847576*x - 3.)
         y2= -.5773502690*x*(-.535898384*x*x - .196152424*x - 1.)
Tile.1 { y1=.5773502690*x*(3.660254040*x^3 - 6.78460970*x*x + 1.85640646*x + 3.)
         y2= -.5773502690*x*(3.660254040*x^3 - 7.85640646*x*x + 3.464101616*x - 1.)
Tile.2 { y1= x*(5*x^3 - 10*x*x + 5*x + 1)
         y2= -x*(5*x^3 - 10*x*x + 5*x - 1)
```



```
#                              TILE PROBLEMS
#                              =============
# PROBLEM 1.0:
# ============
# A manufacturer is designing square floor tiles with unit length of
# side with two curves separating the two colours as shown.
```




```
# The two curves are cubics, positioned so that they trisect the angles
# at the corners.
# Determine the relative amount of the two colours.
# (Try:   "display Tile.Plots"   for graphical display).
#=================================================================
# Assume the origin at the lower left corner with the tile in the first
# quadrant. Define a cubic through (0,0):
> y:=x*(a*x^2+b*x+c);
```

$$y := x\,(a\,x^2 + b\,x + c)$$

```
# Set (x,y)=(1,1) and solve for 'c':
> c:=solve(subs(x=1,")=1,c);
```

$$c := -\,a - b + 1$$

```
# Write the equations for the two curves:
> y1:=subs(a=a1,b=b1,y);
> y2:=subs(a=a2,b=b2,y);
```

$$y1 := x\,(a1\,x^2 + b1\,x - a1 - b1 + 1)$$

$$y2 := x\,(a2\,x^2 + b2\,x - a2 - b2 + 1)$$

```
# Angles of 30 and 60 degrees correspond to slopes of 1/sqrt(3) and
# sqrt(3) respectively. Thus for the 'top' curve, we have
# y'(0)=sqrt(3) and y'(1)=1/sqrt(3). These conditions can be used
# to solve for a1 and b1 say.
> y1p:=diff(y1,x);
> b1:=solve(subs(x=0,")=sqrt(3),b1);
> a1:=solve(subs(x=1,y1p)=1/sqrt(3),a1);
```

$$y1p := a1\,x^2 + b1\,x - a1 - b1 + 1 + x\,(2\,a1\,x + b1)$$

$$b1 := 1 - a1 - 3^{1/2}$$

$$a1 := -\,2 + 3^{1/2} + \frac{1}{3^{1/2}}$$

```
# Similarly for the 'lower' curve ,where we have now y'(0)=1/sqrt(3) and
# y'(1)=sqrt(3):
> y2p:=diff(y2,x);
> b2:=solve(subs(x=0,")=1/sqrt(3),b2);
> a2:=solve(subs(x=1,y2p)=sqrt(3),a2);
```

$$y2p := a2\, x^2 + b2\, x - a2 - b2 + 1 + x\,(2\, a2\, x + b2)$$

$$b2 := 1 - a2 - \frac{1}{3^{1/2}}$$

$$a2 := -2 + 3^{1/2} + \frac{1}{3^{1/2}}$$

```
# Calculate area between curves:
> red:=int(y1-y2,x=0..1);
```

$$red := 1/6\ 3^{1/2} - 1/6\ \frac{1}{3^{1/2}}$$

```
# The blue area is simply the area of the square minus the
# red area. Thus, the ratio of areas is:
> ratio:=factor(red/(1-red));
```

$$ratio := \frac{1}{3\ 3^{1/2} - 1}$$

```
# or approximately:
> evalf(");
```

$$.2383135546$$

```
# The equations of the curves are:
> y1:=factor(y1);
> y2:=factor(y2);
```

$$y1 := -\frac{x\,(2\ 3^{1/2}\, x^2 - 4\, x^2 - 3\ 3^{1/2}\, x + 7\, x - 3)}{3^{1/2}}$$

$$y2 := -\frac{x\,(2\ 3^{1/2}\, x^2 - 4\, x^2 - 3\ 3^{1/2}\, x + 5\, x - 1)}{3^{1/2}}$$

```
# PROBLEM 1.1:
# ===========
# For the square floor tiles of problem 1.0, suppose the two curves
# separate three colours in the upper left,middle,and lower right segments.
# The two curves are quartics, organized so that they trisect the angles
# at the corners, and each colour occupies one-third of the area.
# Determine the equations of the quartics.
#==========================================================================
# Assume the origin at the lower left corner with the tile in the first
# quadrant. Define a quartic through (0,0):
> y:=x*(a*x^3+b*x^2+c*x+d);
                                        3      2
                          y := x (a x  + b x  + c x + d)

------------------------------------------------------------------------------
# Set (x,y)=(1,1) and solve for 'd':
> d:=solve(subs(x=1,")=1,d);
                               d := - a - b - c + 1

------------------------------------------------------------------------------
# Write the equations for the two curves:
> y1:=subs(a=a1,b=b1,c=c1,y);
> y2:=subs(a=a2,b=b2,c=c2,y);
                             3       2
            y1 := x (a1 x  + b1 x  + c1 x - a1 - b1 - c1 + 1)

                             3       2
            y2 := x (a2 x  + b2 x  + c2 x - a2 - b2 - c2 + 1)

------------------------------------------------------------------------------
# Angles of 30 and 60 degrees correspond to slopes of 1/sqrt(3) and
# sqrt(3) respectively. Thus for the 'top' curve, we have
# y'(0)=sqrt(3) and y'(1)=1/sqrt(3). These conditions can be used
# to solve for b1 and c1 say.
> y1p:=diff(y1,x);
> c1:=solve(subs(x=0,")=sqrt(3),c1);
> b1:=solve(subs(x=1,y1p)=1/sqrt(3),b1);
             3       2                                          2
  y1p := a1 x  + b1 x  + c1 x - a1 - b1 - c1 + 1 + x (3 a1 x  + 2 b1 x + c1)

                                                  1/2
                          c1 := 1 - a1 - b1 - 3

                                            1/2     1
                      b1 := - 2 a1 - 2 + 3    + ------
                                                    1/2
                                                   3

------------------------------------------------------------------------------
# Similarly for the 'lower' curve:
> y2p:=diff(y2,x);
> c2:=solve(subs(x=0,")=1/sqrt(3),c2);
> b2:=solve(subs(x=1,y2p)=sqrt(3),b2);
             3       2                                          2
  y2p := a2 x  + b2 x  + c2 x - a2 - b2 - c2 + 1 + x (3 a2 x  + 2 b2 x + c2)

                                                1
                         c2 := 1 - a2 - b2 - ------
                                                 1/2
                                                3

                                            1        1/2
                      b2 := - 2 a2 - 2 + ------ + 3
                                             1/2
                                            3
```

```
# The area under the 'lower' curve is to be 1/3, and this condition
# can be used to determine the value of a2:
> area1:=int(y2,x=0..1);
> a2:=solve("=1/3,a2);
```

$$area1 := 1/30\ a2 + 1/2 + 1/12 \frac{1}{3^{1/2}} - 1/12\ 3^{1/2}$$

$$a2 := -5 - 5/2 \frac{1}{3^{1/2}} + 5/2\ 3^{1/2}$$

```
# The area between the curves is also 1/3. This allows the evaluation of
# a1:
> area2:=int(y1-y2,x=0..1);
> a1:=solve("=1/3,a1);
```

$$area2 := 1/30\ a1 + 1/6 + 1/12\ 3^{1/2} - 1/12 \frac{1}{3^{1/2}}$$

$$a1 := 5 + 5/2 \frac{1}{3^{1/2}} - 5/2\ 3^{1/2}$$

```
# The equations of the curves are:
> y1:=factor(y1);
> y2:=factor(y2);
> y1:=evalf(y1);
> y2:=evalf(y2);
```

$$y1 := \frac{x\ (5\ 3^{1/2}\ x^3 - 5\ x^3 - 12\ 3^{1/2}\ x^2 + 14\ x^2 + 8\ 3^{1/2}\ x - 12\ x + 3)}{3^{1/2}}$$

$$y2 := -\frac{x\ (5\ 3^{1/2}\ x^3 - 5\ x^3 - 8\ 3^{1/2}\ x^2 + 6\ x^2 + 2\ 3^{1/2}\ x - 1)}{3^{1/2}}$$

$$y1 := .5773502690\ x\ (3.660254040\ x^3 - 6.78460970\ x^2 + 1.85640646\ x + 3.)$$

$$y2 := -\ .5773502690\ x\ (3.660254040\ x^3 - 7.85640646\ x^2 + 3.464101616\ x - 1.)$$

```
# PROBLEM 1.2:
# ===========
# For the square floor tiles of problem 1.0, suppose the two curves
# separate three colours in the upper left,middle,and lower right segments.
# The two curves are quartics, arranged so that their slopes at both corners
# is one, and each colour occupies one-third of the area.
# Determine the equations of the quartics.
#=========================================================================
# Assume the origin at the lower left corner with the tile in the first
# quadrant. Since both curves have similar constraints on their shape,we
# can determine their common coefficients first.
# Define a quartic through (0,0):
> y:=x*(a*x^3+b*x^2+c*x+d);
                                     3      2
                         y := x (a x  + b x  + c x + d)
-------------------------------------------------------------------------------
# Set (x,y)=(1,1) and solve for 'd':
> d:=solve(subs(x=1,")=1,d);
                              d := - a - b - c + 1
-------------------------------------------------------------------------------
# The unit slope at [0,0] and [1,1] for the two curves fix two other
# coefficients. That is, y'(0)=1 and y'(1)=1 can be used to solve
# for 'b' and 'c' say.
> yp:=diff(y,x);
> c:=solve(subs(x=0,")=1,c);
> b:=solve(subs(x=1,yp)=1,b);
              3      2                                       2
     yp := a x  + b x  + c x - a - b - c + 1 + x (3 a x  + 2 b x + c)

                                  c := - a - b

                                   b := - 2 a
-------------------------------------------------------------------------------
# Now two curves which satisfy the "shape" conditions are:
> y1:=subs(a=a1,y);
> y2:=subs(a=a2,y);
                                 3         2
                     y1 := x (a1 x  - 2 a1 x  + a1 x + 1)

                                 3         2
                     y2 := x (a2 x  - 2 a2 x  + a2 x + 1)
-------------------------------------------------------------------------------
# The area under the 'lower' curve is to be 1/3, and this condition
# can be used to determine the value of 'a2':
> area1:=int(y2,x=0..1);
> a2:=solve("=1/3,a2);
                            area1 := 1/30 a2 + 1/2

                                   a2 := -5
-------------------------------------------------------------------------------
# The area between the curves is also 1/3. This allows the evaluation of 'a1':
> area2:=int(y1-y2,x=0..1);
> a1:=solve("=1/3,a1);
                            area2 := 1/30 a1 + 1/6

                                   a1 := 5
-------------------------------------------------------------------------------
# The equations of the curves are:
> y1:=factor(y1);
> y2:=factor(y2);
                                   3       2
                       y1 := x (5 x  - 10 x  + 5 x + 1)

                                     3       2
                      y2 := - x (5 x  - 10 x  + 5 x - 1)
-------------------------------------------------------------------------------
```



Problem 1.0

Wall cross sections for n=2, 3, 4, 5, 10 and 20

Sequence of wall cross sections is given by

$$y = \frac{5}{3^{n+1}} (3-x)^n (nx+3)$$

and converge to a vertical line segment of height 5 at x=0, with y=0 for 0 < x <= 3.



Problem 1.5

CONCRETE WALL PROBLEMS

```
# PROBLEM 1.0:
# ===========
# A 5 foot circular concrete wall, 3 feet wide at the base, is to be built
# around a circular swimmimg pool. The inner radius of the wall is 30 feet.
# Assuming that the wall begins and ends on the ground and that the
# cross section of the wall is a cubic polynomial determine that cubic
# which minimizes the volume of concrete used.
# ==========================================================================
# Let y=0 be the base level and x=0 be the location of the inner radius.
# Define cubic through (0,0) and (3,0):
> y:=x*(3-x)*(a*x+b);
```

$$y := x\,(3 - x)\,(a\,x + b)$$

```
# Compute derivative  Dy = dy/dx:
> Dy:=diff(y,x);
```

$$Dy := (3 - x)\,(a\,x + b) - x\,(a\,x + b) + x\,(3 - x)\,a$$

```
# Expand above expression:
> Dy:=expand(Dy);
```

$$Dy := 6\,a\,x + 3\,b - 3\,a\,x^2 - 2\,x\,b$$

```
# Solve equations  y=5, Dy=0  (at the extreme point) for "a" and "b":
> solve ( {y=5,Dy=0},{a,b});
```

$$\left\{a = 5\,\frac{-3 + 2\,x}{x^2\,(9 - 6\,x + x^2)},\ b = -15\,\frac{-2 + x}{x\,(9 - 6\,x + x^2)}\right\}$$

```
# Let x=M be the location of the maxmum. Substitute into "a", "b" and simplify:
> ab:=factor( subs(x=M,") );
```

$$ab := \left\{a = 5\,\frac{-3 + 2\,M}{M^2\,(M - 3)^2},\ b = -15\,\frac{-2 + M}{M\,(M - 3)^2}\right\}$$

```
# Substitute a(M), b(M) into cubic (giving 1-parameter family of cubics):
> y:=factor( subs(ab,y) );
```

$$y := -5\,\frac{x\,(-3 + x)\,(-3\,x + 2\,M\,x + 6\,M - 3\,M^2)}{M^2\,(M - 3)^2}$$

```
# Substitute a(M), b(M) into Dy=dy/dx:
> Dy:=factor( subs(ab,Dy) );
```

$$Dy := -15\,\frac{(-3\,x + 2\,M\,x + 6 - 3\,M)\,(-M + x)}{M^2\,(M - 3)^2}$$

```
# Determine Dy=dy/dx at x=0:
> subs(x=0,Dy);
```

$$15\,\frac{6 - 3\,M}{M\,(M - 3)^2}$$

```
# Requiring Dy=dy/dx >= 0 at x=0 requires that M <= 2.
# Now determine Dy=dy/dx at x=3:
> subs(x=3,Dy);
```

$$- 15 \frac{(- 3 + 3 M) (- M + 3)}{M^2 (M - 3)^2}$$

```
# Requiring Dy=dy/dx <= 0 at x=3 requires that M >= 1.
# From above two conditions we see that 1 <= M <= 2.
# Now evaluate volume of solid of revolution:
> V:=factor( int(2*Pi*(x+30)*y,x=0..3) );
```

$$V := - 81/2 \frac{Pi (159 - 316 M + 105 M^2)}{M^2 (M - 3)^2}$$

```
# Problem now is to maximize V(M) for 1 <= M <= 2
# Calculate DV = dV/dM (where M is location of maximum wall height)
> DV:=factor( diff(V,M) );
```

$$DV := 243 \frac{Pi (35 M - 53) (M^2 - 3 M + 3)}{M^3 (M - 3)^3}$$

```
# Determine critical points (in 1 < M < 2), where dV/dM = 0:
> solve (DV=0,M);
```

$$\frac{53}{35}, 3/2 + 1/2 I 3^{1/2}, 3/2 - 1/2 I 3^{1/2}$$

```
# The only real root is M = 53/35.
# Evaluate V at M=1, M=2 (end points) and at M=53/35 (critical point):
> evalf( subs(M=1,V) );
> evalf( subs(M=2,V) );
> evalf( subs(M=53/35,V) );
```

$$1654.048532$$

$$1685.857158$$

$$1979.382908$$

```
# The minimum volume V occurs when the maximum wall height occurs at x=M=1.
# Determine the optimal cubic by substituting M=1 into 1-parameter family:
> y:=factor( subs(M=1,y) );
```

$$y := 5/4 x (- 3 + x)^2$$

```
# PROBLEM 1.1:
# ===========
# For the concrete wall of problem 1.0 assume that the cross section of
# the wall is an (n+2)nd degree polynomial with a factor x^2 (3-x)^n.
# Find all such polynomials whose maximum height is 5 feet (for x in [0,3] ),
# and determine the volume of concrete required for n=2, 3, 4, 5, 10 and 20.
# ===========================================================================
# Define polynomial:
> y:=a*x^2*(3-x)^n;
                                       2        n
                              y := a x  (3 - x)
--------------------------------------------------------------------------------
# Compute derivative  Dy = dy/dx:
> Dy:=factor( diff(y,x) );
                                       n
                           a x (3 - x)  (- 6 + 2 x + x n)
                   Dy := ------------------------------
                                      - 3 + x
--------------------------------------------------------------------------------
# Solve Dy=dy/dx=0 for x=M (location of maximum height):
> solve(Dy=0,x);
                                           1
                               0, 3, 6 -------
                                         2 + n
--------------------------------------------------------------------------------
# The location, in (0,3), is:
> M:=6/(2+n);
                                           1
                                M := 6 -------
                                         2 + n
--------------------------------------------------------------------------------
# Substitute x=M into y:
> subs( x=M,y );
                                        1     n
                              a (3 - 6 -------)
                                        2 + n
                          36 -------------------
                                           2
                                    (2 + n)
--------------------------------------------------------------------------------
# Set y=5 at x=M  and solve for "a":
> a:=solve("=5,a);
                                                2
                                         (2 + n)
                        a := 5/36 -----------------
                                              1     n
                                     (3 - 6 -------)
                                              2 + n
--------------------------------------------------------------------------------
# We now have all such polynomials:
> y;
                                        2  2        n
                                 (2 + n)  x  (3 - x)
                        5/36 ----------------------
                                          1     n
                                 (3 - 6 -------)
                                          2 + n

--------------------------------------------------------------------------------
# Set n=2, 3, ... , find y and the volume of concrete:
> Digits:=4:
--------------------------------------------------------------------------------
# For n=2:
> y2:=factor( subs(n=2,y) );
> V:=evalf( int(2*Pi*(x+30)*y2,x=0..3) );
```

$$y2 := \frac{80}{81} x^2 (-3 + x)^2$$

$$V := 1584.$$

```
# For n=3:
> y3:=factor( subs(n=3,y) );
> V:=evalf( int(2*Pi*(x+30)*y3,x=0..3) );
```

$$y3 := -\frac{15625}{26244} x^2 (-3 + x)^3$$

$$V := 1422.$$

```
# For n=4:
> y4:=factor( subs(n=4,y) );
> V:=evalf( int(2*Pi*(x+30)*y4,x=0..3) );
```

$$y4 := 5/16\, x^2 (-3 + x)^4$$

$$V := 1273.$$

```
# For n=5:
> y5:=factor( subs(n=5,y) );
> V:=evalf( int(2*Pi*(x+30)*y5,x=0..3) );
```

$$y5 := -\frac{823543}{5467500} x^2 (-3 + x)^5$$

$$V := 1146.$$

```
# For n=10:
> y10:=factor( subs(n=10,y) );
> V:=evalf( int(2*Pi*(x+30)*y10,x=0..3) );
```

$$y10 := \frac{4096}{1953125} x^2 (-3 + x)^{10}$$

$$V := 750.3$$

```
# For n=20:
> y20:=factor( subs(n=20,y) );
> V:=evalf( int(2*Pi*(x+30)*y20,x=0..3) );
```

$$y20 := \frac{81402749386839761113321}{6276211921800000000000000000} x^2 (-3 + x)^{20}$$

$$V := 438.6$$

```
# For n=30:
> y30:=factor( subs(n=30,y) );
> V:=evalf( int(2*Pi*(x+30)*y30,x=0..3) );
```

$$y30 := \frac{340282366920938463463374607431768211456}{710637167984518840217310746775940060615539550781 25} x^2 (-3 + x)^{30}$$

$$V := 309.3$$

```
# PROBLEM 1.2:
# ==========
# For the  circular concrete wall of problem 1.0, assume that the inside face
# of the wall is 5 feet high and that the top of the wall is horizontal at this
# point. Assuming that the cross section of the wall is a cubic polynomial,
# determine that cubic which minimizes the volume of concrete used.
# ======================================================================
# Let y=0 be the base level and x=0 be the location of the inner radius.
# Define cubic through (3,0):
> y:=(3-x)*(a*x^2+b*x+c);
                                               2
                         y := (3 - x)  (a x  + b x + c)

--------------------------------------------------------------------------------
# We substitute x=0 in y:
> subs(x=0,y);
                                    3 c

--------------------------------------------------------------------------------
# Set y=5 and solve for "c":
> c:=solve( "=5,c );
                                  c := 5/3

--------------------------------------------------------------------------------
# Compute derivative  Dy = dy/dx:
> Dy:=expand( diff(y,x) );
                              2
                 Dy := - 3 a x  - 2 b x - 5/3 + 6 a x + 3 b

--------------------------------------------------------------------------------
# We substitute x=0 in Dy=dy/dx:
> subs(x=0,Dy);
                                - 5/3 + 3 b

--------------------------------------------------------------------------------
# Set Dy=dy/dx=0 and solve for "b":
> b:=solve ( "=0,b );
                                  b := 5/9

--------------------------------------------------------------------------------
# We now have all cubics with y=5, dy/dx=0 at x=0 AND y=0 when x=3.
> y;
                                      2
                         (3 - x)  (a x  + 5/9 x + 5/3)

--------------------------------------------------------------------------------
# Determine Dy=dy/dx at x=3:
> subs(x=3,Dy);
                                - 9 a - 10/3

--------------------------------------------------------------------------------
# Requiring Dy=dy/dx <= 0 at x=3 requires that a >= -10/27.
# Now evaluate volume of solid of revolution, V(a):
> V:=factor( int(2*Pi*(x+30)*y,x=0..3) );
                          V := 3/10 Pi  (1431 a + 2075)

--------------------------------------------------------------------------------
# ************************************************************
# *    Problem: maximize V(a) on the interval  a >= -10/27    *
# ************************************************************
# Calculate DV = dV/da:
> DV:=factor( diff(V,a) );
                                    4293
                              DV := ------ Pi
                                     10
```

```
---------------------------------------------------------------------------
# Since DV=dV/da>0, minimum V occurs at minimum "a", namely a = -10/27.
> a:=-10/27;
                                           10
                                  a := - ----
                                           27

---------------------------------------------------------------------------
# The minimal volume of concrete is:
> evalf(V);
                                  1456.128195

---------------------------------------------------------------------------
# The optimal cubic is:
> y:=factor(y);
                                            2
                          y := 5/27 (x - 3)  (2 x + 3)

---------------------------------------------------------------------------
```

```
# PROBLEM 1.3:
# ===========
# For the circular concrete wall of problem 1.2, assume that the cross section
# of the wall is a 4th degree polynomial. Determine that quartic which minimizes
# the volume of concrete used.
# ===========================================================================
# Let y=0 be the base level and x=0 be the location of the inner radius.
# Define quartic through (3,0):
> y:=(3-x)*(a*x^3+b*x^2+c*x+d);
                                         3      2
                        y := (3 - x)  (a x  + b x  + c x + d)

--------------------------------------------------------------------------------
# We substitute x=0 in y:
> subs(x=0,y);
                                       3 d

--------------------------------------------------------------------------------
# Set y=5 and solve for "d":
> d:=solve( "=5,d );
                                    d := 5/3

--------------------------------------------------------------------------------
# Compute derivative  Dy = dy/dx:
> Dy:=expand( diff(y,x) );
                     3        2                          2
        Dy := - 4 a x  - 3 b x  - 2 c x - 5/3 + 9 a x  + 6 b x + 3 c

--------------------------------------------------------------------------------
# We substitute x=0 in Dy=dy/dx:
> subs(x=0,Dy);
                                  - 5/3 + 3 c

--------------------------------------------------------------------------------
# Set Dy=dy/dx=0 and solve for "c":
> c:=solve ( "=0,c );
                                    c := 5/9

--------------------------------------------------------------------------------
# We now have all quartics with y=5, dy/dx=0 at x=0 AND y=0 when x=3.
> y;
                                    3      2
                        (3 - x)  (a x  + b x  + 5/9 x + 5/3)

--------------------------------------------------------------------------------
# The derivatives of such quartics are:
> Dy;
                         3        2                    2
                  - 4 a x  - 3 b x  - 10/9 x + 9 a x  + 6 b x

--------------------------------------------------------------------------------
# We substitute x=3 in Dy;
> subs(x=3,Dy);
                              - 27 a - 9 b - 10/3

--------------------------------------------------------------------------------
# We require Dy=dy/dx<=0 at x=3. This requires b>=B(a).
> B:=solve("=0,b);
                                              10
                                B := - 3 a - ----
                                              27

--------------------------------------------------------------------------------
# Now evaluate volume of solid of revolution, V(a,b):
> V:=factor( int(2*Pi*(x+30)*y,x=0..3) );
                     V := 3/10 Pi  (2592 a + 1431 b + 2075)
```

```
---------------------------------------------------------------------------
# Note that V(a,b) increases with "b".
# Hence, for any given "a", we choose the minimum "b", namely b=B(a):
> b:=B;
                                         10
                             b := - 3 a - ----
                                         27

---------------------------------------------------------------------------
# The quartic is now:
> y;
                    3                 10    2
       (3 - x) (a x  + (- 3 a - ----) x  + 5/9 x + 5/3)
                                 27

---------------------------------------------------------------------------
# The volume of concrete is then V=V(a):
> V;
                      3/10 Pi (- 1701 a + 1545)

---------------------------------------------------------------------------
# (Note that V(a) decreases with "a", so we must choose the MAXIMUM "a").
# Now our quartic has a zero slope at x=3. To guarantee that y>=0 in (0,3)
# we require that the 2nd derivative of y is >=0 at x=3.
# First we compute the 2nd derivative:
> DDY:=factor( diff(Dy,x) );
                            2
           DDY := - 12 a x  + 36 a x + 20/9 x - 10/3 - 18 a

---------------------------------------------------------------------------
# Then substitute x=3:
> subs(x=3,");
                                 - 18 a + 10/3

---------------------------------------------------------------------------
# In order to guarantee "2nd derivative>=0", we need a<=A:
> A:=solve("=0,a);
                                  A := 5/27

---------------------------------------------------------------------------
# Substituing this MAXIMUM "a" gives the minimum volume of concrete:
> a:=A;
> evalf(V);
                                  a := 5/27

                                1159.247689

---------------------------------------------------------------------------
# The optimal quartic is then:
> factor(y);
                                        3
                        - 5/27 (- 3 + x)  (x + 1)

---------------------------------------------------------------------------
```

```
# PROBLEM 1.4:
# ===========
# For the circular concrete wall of problem 1.2, assume that the cross section
# of the wall is a 5th degree polynomial. Determine that quintic which minimizes
# the volume of concrete used.
# ======================================================================
# Define quintic through (3,0):
> y:=(3-x)*(a*x^4+b*x^3+c*x^2+d*x+e);
                                     4      3      2
                 y := (3 - x) (a x  + b x  + c x  + d x + e)
------------------------------------------------------------------------
# We substitute x=0 in y:
> subs(x=0,y);
                                 3 e

------------------------------------------------------------------------
# Set y=5 and solve for "e":
> e:=solve( "=5,e );
                               e := 5/3

------------------------------------------------------------------------
# Compute derivative  Dy = dy/dx:
> Dy:=expand( diff(y,x) );
              4        3        2                           3        2
Dy := - 5 a x  - 4 b x  - 3 c x  - 2 d x - 5/3 + 12 a x  + 9 b x  + 6 c x + 3 d



------------------------------------------------------------------------
# We substitute x=0 in Dy=dy/dx:
> subs(x=0,Dy);
                             - 5/3 + 3 d

------------------------------------------------------------------------
# Set Dy=dy/dx=0 and solve for "d":
> d:=solve ( "=0,d );
                               d := 5/9

------------------------------------------------------------------------
# We now have all quintics with y=5, dy/dx=0 at x=0 AND y=0 when x=3.
> y;
                               4      3      2
                 (3 - x) (a x  + b x  + c x  + 5/9 x + 5/3)

------------------------------------------------------------------------
# The derivatives of such quintics are:
> Dy;
             4        3        2                        3        2
      - 5 a x  - 4 b x  - 3 c x  - 10/9 x + 12 a x  + 9 b x  + 6 c x

------------------------------------------------------------------------
# We substitute x=3 in Dy;
> subs(x=3,Dy);
                      - 81 a - 27 b - 9 c - 10/3

------------------------------------------------------------------------
```

```
# We require Dy=dy/dx<=0 at x=3. This requires c>=C(a,b).
> C:=solve("=0,c);
                                                   10
                         C := - 9 a - 3 b - ----
                                                   27
--------------------------------------------------------------------------------
# Now evaluate volume of solid of revolution, V(a,b,c):
> V:=factor( int(2*Pi*(x+30)*y,x=0..3) );
             V := 3/70 Pi (36450 a + 18144 b + 10017 c + 14525)
--------------------------------------------------------------------------------
# Note that V(a,b,c) increases with "c".
# Hence, for any given (a,b) we choose the minimum "c", namely c=C(a,b):
> c:=C;
                                                   10
                         c := - 9 a - 3 b - ----
                                                   27
--------------------------------------------------------------------------------
# The quintic is now:
> y;
                     4          3                            10       2
    (3 - x)  (a x  + b x  + (- 9 a - 3 b - ----) x  + 5/9 x + 5/3)
                                                             27
--------------------------------------------------------------------------------
# The volume of concrete is then V=V(a,b)):
> V;
                    3/70 Pi (- 53703 a - 11907 b + 10815)
--------------------------------------------------------------------------------
# ( Note that V decreases "b" ).
# Now our quintic has a zero slope at x=3. To guarantee that y>=0 in (0,3)
# we require that the 2nd derivative of y is >=0 at x=3.
# First we compute the 2nd derivative:
> DDY:=factor( diff(Dy,x) );
                     3            2                                                    2
DDY := - 20 a x  - 12 b x  + 54 a x + 36 b x + 20/9 x - 10/3 + 36 a x  - 54 a

    - 18 b
--------------------------------------------------------------------------------
# Then substitute x=3:
> subs(x=3,");
                          - 108 a - 18 b + 10/3
--------------------------------------------------------------------------------
# In order to guarantee "2nd derivative>=0", we need b<=B:
> B:=solve("=0,b);
                            B := - 6 a + 5/27
--------------------------------------------------------------------------------
# Substituing this MAXIMUM "b" gives the volume of concrete:
> b:=B;
> V;
                            b := - 6 a + 5/27

                         3/70 Pi (17739 a + 8610)
--------------------------------------------------------------------------------
# (Note that the volume increases with "a", so we need the MAXIMUM "a").
# Now our quintic has y=0, Dy=0 AND DDY=0 at x=3. In order that y>=0 in (0,3)
# we require that the 3rd derivative of y is >=0 at x=3.
# First we compute the 3rd derivative:
> DDDy:=diff( DDY,x );
```

```
                                     2
        DDDy := - 60 a x  - 24 (- 6 a + 5/27) x - 162 a + 80/9 + 72 a x

---------------------------------------------------------------------------
# Then substitute x=3;
> subs( x=3, DDDy );
                              - 54 a - 40/9

---------------------------------------------------------------------------
# In order to satisfy "3rd derivative >=0 at x=3" we need a<=A:
> A:=solve( "=0,a);
                                       20
                               A := - -----
                                       243

---------------------------------------------------------------------------
# Substutute a=A to obtain the minimal volume of concrete:
> a:=A;
> evalf(V);
                                       20
                               a := - -----
                                       243

                               962.6737489

---------------------------------------------------------------------------
# The optimal quintic is then:
> factor(y);
                                      4
                          5/243 (x - 3)  (4 x + 3)

---------------------------------------------------------------------------
```

PROBLEM 1.5:

```
# ==========
# For the circular concrete wall of problem 1.2, assume that the cross section
# is given by an (n+1)st degree polynomial with a factor (3-x)^n.
# Find all polynomials which satisfy the given constraints (At x=0: y=5 and
# dy/dx=0. At x=3: y=0).
# Find the volume of concrete used for n=2, 3, 4, 5, 10 and 20.
# Plot the polynomials for each n.
# What is the limiting curve as n->infinity?
# What is the limiting volume of concrete?
# (Try:   "display Wall.5.Plots"   for graphical display).
# ==========================================================================
# Let y=0 be the base level and x=0 be the location of the inner radius.
# Define polynomial of (n+1)st degree;
> y:=(a*x+b)*(3-x)^n;
```

$$y := (a\,x + b)\,(3 - x)^n$$

```
# We substitute x=0 in y:
> subs(x=0,y);
```

$$b\,3^n$$

```
# Set y=5 and solve for "b":
> b:=solve( "=5,b );
```

$$b := 5\,\frac{1}{3^n}$$

```
# Now compute Dy=dy/dx:
> Dy:=diff(y,x);
```

$$Dy := a\,(3 - x)^n - \frac{\left(a\,x + 5\,\frac{1}{3^n}\right)(3 - x)^n\,n}{3 - x}$$

```
# Now subxtitute x=0:
> subs(x=0,");
```

$$a\,3^n - 5/3\,n$$

```
# Solve for "a":
> a:=solve("=0,a);
```

$$a := 5/3\,\frac{n}{3^n}$$

```
# We now have all the polynomials with y=5, dy/dx=0 at x=0 AND y=0 when x=3.
> y:=factor(y);
```

$$y := 5/3\,\frac{(3 - x)^n\,(n\,x + 3)}{3^n}$$

Substutute n=2, 3, 4, ..., determine yn(x), then the volume of concrete:
```
> Digits:=4:
```

For n=2 (cubic cross-section):
```
> y2:=factor( subs(n=2,y) );
> V:=evalf(int(2*Pi*(x+30)*y2,x=0..3));
```
$$y2 := 5/27\ (-3 + x)^2\ (2x + 3)$$

$$V := 1456.$$

For n=3 (quartic cross-section):
```
> y3:=factor( subs(n=3,y) );
> V:=evalf(int(2*Pi*(x+30)*y3,x=0..3));
```
$$y3 := -5/27\ (-3 + x)^3\ (x + 1)$$

$$V := 1159.$$

For n=4:
```
> y4:=factor( subs(n=4,y) );
> V:=evalf( int(2*Pi*(x+30)*y4,x=0..3) );
```
$$y4 := 5/243\ (-3 + x)^4\ (4x + 3)$$

$$V := 962.7$$

For n=5:
```
> y5:=factor( subs(n=5,y) );
> V:=evalf( int(2*Pi*(x+30)*y5,x=0..3) );
```
$$y5 := -5/729\ (-3 + x)^5\ (5x + 3)$$

$$V := 823.2$$

For n=10:
```
> y10:=factor( subs(n=6,y) );
> V:=evalf( int(2*Pi*(x+30)*y10,x=0..3) );
```
$$y10 := 5/729\ (-3 + x)^6\ (2x + 1)$$

$$V := 718.9$$

For n=20:
```
> y20:=factor( subs(n=20,y) );
> V:=evalf( int(2*Pi*(x+30)*y20,x=0..3) );
```
$$y20 := 5/10460353203\ (-3 + x)^{20}\ (20x + 3)$$

$$V := 258.7$$

For n=30:
```
> y30:=factor( subs(n=30,y) );
> V:=evalf( int(2*Pi*(x+30)*y30,x=0..3) );
```
$$y30 := 5/205891132094649\ (-3 + x)^{30}\ (10x + 1)$$

$$V := 177.6$$
